Chapter 1b

Predictive Analytics I- Exponential Smoothing

Exponential Smoothing (ES) is another forecasting technique where the forecast for the next period is the same as the forecast for this period plus a fraction of the gap between the actual and the forecast of this period. The fraction is defined by coefficient α, which is between 0 and 1. Suppose we are in period 3, suppose A3=10, F3=12, and α=0.5. Then our forecast for period 4 is F4 = F3+ α(A3-F3) ➔ F4 = 12+0.5(10-12) ➔ F4=11.

In general, the exponential smoothing formula is stated as

$$\mathbf{F_{t+1} = F_t + \alpha (A_t - F_t)}$$

The forecast for the first period is usually set to the actual for the first period.

We can manipulate the formula and write it in a different form

$F_{t+1} = F_t + \alpha (A_t - F_t)$

$F_{t+1} = F_t + \alpha A_t - \alpha F_t$

$$\mathbf{F_{t+1} = (1-\alpha)F_t + \alpha A_t}$$

We usually use the latter.

Exponential Smoothing is a weighted moving average where all the past data are present. The weight of data decreases as their age increases.

When comparing the moving average with a weighted moving average, there is a distinct difference between the two methods.

- In a moving average, the periods are all weighted equally and therefore given equal importance.
- In a weighted moving average, there is more "weight" given to the most recent data and less "weight" given to older data. The equations below clearly show these differences.

4-Period-Moving-Average (4-Period)

$F_{t+1} = (A_t + A_{t-1} + A_{t-2} + A_{t-3})/4$

We can write it as

$F_{t+1} = A_t/4 + A_{t-1}/4 + A_{t-2}/4 + A_{t-3}/4$

In other words

$F_{t+1}= 0.25A_t + 0.25A_{t-1} + 0.25A_{t-2} + 0.25A_{t-3}$

Therefore, a moving average is also a weighted moving average where the weights are equal as opposed to an actual weighted moving average which may look like

$F_{t+1}= 0.35A_t + 0.30A_{t-1} + 0.25A_{t-2} + 0.1A_{t-3}$

Note that the sum of all coefficients for each technique is equal to 1.

Example: Raw Materials for Production. In order to meet market demand for a rare disease medicine, the company Drugs-R-Us needs to purchase enough raw materials to support production requirements. However, the Warehouse Manager wants to keep inventory as low as possible while still supporting production. This will minimize tax expenses from storing surplus inventory at the end of each month. The table below shows the forecast for material purchases using the exponential smoothing method with an alpha of 0.4. It is noteworthy to mention that the forecast in period 2 was obtained using the Naive Method.

	α=	0.4					
t	At	Ft		|E|=|At-Ft|		MAD	
1	130	130.0	=B3	0.00	=ABS(B3-C3)	0.00	=AVERAGE($E$3:E3)
2	150	130.0	=(1-$C$1)*C3+$C$1*B3	20.00	=ABS(B4-C4)	10.00	=AVERAGE($E$3:E4)
3	80	138.0	=(1-$C$1)*C4+$C$1*B4	58.00	=ABS(B5-C5)	26.00	=AVERAGE($E$3:E5)
4	190	114.8	=(1-$C$1)*C5+$C$1*B5	75.20	=ABS(B6-C6)	38.30	=AVERAGE($E$3:E6)
5	140	144.9	=(1-$C$1)*C6+$C$1*B6	4.88	=ABS(B7-C7)	31.62	=AVERAGE($E$3:E7)
6	150	142.9	=(1-$C$1)*C7+$C$1*B7	7.07	=ABS(B8-C8)	27.53	=AVERAGE($E$3:E8)
7	120	145.8	=(1-$C$1)*C8+$C$1*B8	25.76	=ABS(B9-C9)	27.27	=AVERAGE($E$3:E9)
8	140	135.5	=(1-$C$1)*C9+$C$1*B9	4.55	=ABS(B10-C10)	24.43	=AVERAGE($E$3:E10)
9	100	137.3	=(1-$C$1)*C10+$C$1*B10	37.27	=ABS(B11-C11)	25.86	=AVERAGE($E$3:E11)
10	60	122.4	=(1-$C$1)*C11+$C$1*B11	62.36	=ABS(B12-C12)	29.51	=AVERAGE($E$3:E12)
11	11	97.4	=(1-$C$1)*C12+$C$1*B12	86.42	=ABS(B13-C13)	34.68	=AVERAGE($E$3:E13)
12	150	62.9	=(1-$C$1)*C13+$C$1*B13	87.15	=ABS(B14-C14)	39.05	=AVERAGE($E$3:E14)
13		97.7	=(1-$C$1)*C14+$C$1*B14				
Forecast for next period:				**μ=**	**97.7**	**σ=**	**48.8**
					=C15		=1.25*G14

Another forecast was generated for comparison. The exponential smoothing results below were generated using a **α**-value of 0.8.

	α=	0.8					
t	At	Ft		|E|=|At-Ft|		MAD	
1	130	130.0	=B3	0.00	=ABS(B3-C3)	0.00	=AVERAGE($E$3:E3)
2	150	130.0	=(1-$C$1)*C3+$C$1*B3	20.00	=ABS(B4-C4)	10.00	=AVERAGE($E$3:E4)
3	80	146.0	=(1-$C$1)*C4+$C$1*B4	66.00	=ABS(B5-C5)	28.67	=AVERAGE($E$3:E5)
4	190	93.2	=(1-$C$1)*C5+$C$1*B5	96.80	=ABS(B6-C6)	45.70	=AVERAGE($E$3:E6)
5	140	170.6	=(1-$C$1)*C6+$C$1*B6	30.64	=ABS(B7-C7)	42.69	=AVERAGE($E$3:E7)
6	150	146.1	=(1-$C$1)*C7+$C$1*B7	3.87	=ABS(B8-C8)	36.22	=AVERAGE($E$3:E8)
7	120	149.2	=(1-$C$1)*C8+$C$1*B8	29.23	=ABS(B9-C9)	35.22	=AVERAGE($E$3:E9)
8	140	125.8	=(1-$C$1)*C9+$C$1*B9	14.15	=ABS(B10-C10)	32.59	=AVERAGE($E$3:E10)
9	100	137.2	=(1-$C$1)*C10+$C$1*B10	37.17	=ABS(B11-C11)	33.10	=AVERAGE($E$3:E11)
10	60	107.4	=(1-$C$1)*C11+$C$1*B11	47.43	=ABS(B12-C12)	34.53	=AVERAGE($E$3:E12)
11	11	69.5	=(1-$C$1)*C12+$C$1*B12	58.49	=ABS(B13-C13)	36.71	=AVERAGE($E$3:E13)
12	150	22.7	=(1-$C$1)*C13+$C$1*B13	127.30	=ABS(B14-C14)	44.26	=AVERAGE($E$3:E14)
13		124.5	=(1-$C$1)*C14+$C$1*B14				
Forecast for next period:				**μ=**	**124.5**	**σ=**	**55.3**
					=C15		=1.25*G14

Since the MAD value for **α** = 0.4 is smaller than the **α** = 0.8, the better forecast is the forecast with a smaller alpha. This is not a general rule, that means we do not know whether smaller or larger **α** is the best fit for our data. However, it is a general rule that the smaller **α**-value is less reactive and is better for long-range planning purposes. The larger **α**-value is more reactive and best utilized for short-range forecasts. Indeed, if we set **α** = 0, that means we pay no attention to reality, and stay with an original forecast. In contrast, **α** = 1 means that we put our full trust in the most recent piece of data.

Question **α** = 1 is the same as what moving average? One-period moving average.

These data sets were generated using 2 **α**-values, and it was determined **α** = 0.4 is better than **α** = 0.8. To determine the best **α**-value for the best exponential smoothing forecast, the following table was generated using the "What-If-Analysis" option in Excel and the starting with α =0.4, and MAD = 39.

Type 0 in one cell and 0.1 in the cell below it.

0				
0.1				
	44.26			
0			=J1	
0.1			=J5+$J$2	
0.2			=J6+$J$2	
0.3			=J7+$J$2	
0.4			=J8+$J$2	
0.5			=J9+$J$2	
0.6			=J10+$J$2	
0.7			=J11+$J$2	
0.8			=J12+$J$2	
0.9			=J13+$J$2	
1			=J14+$J$2	

J	K	L	M
0			
0.1			
	44.26		
0			=J1
0.1			=J5+$J$2
0.2			=J6+$J$2
0.3			=J7+$J$2
0.4			=J8+$J$2
0.5			=J9+$J$2
0.6			=J10+$J$2
0.7			=J11+$J$2
0.8			=J12+$J$2
0.9			=J13+$J$2
1			=J14+$J$2

Mark from J3 to K14. That is a table with two columns and 12 rows where current MAD is in K4, and potential α values are typed in column J from J4 to J14.

Go to what-if analysis, and click on the data table.



Your data is not typed in a row; they are typed in a column. Put your mouse in “Column input cell.” After having the mouse in “Column input cell,” click on the cell containing the α value. The table is automatically filled.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1		α=	0.4							0				
2	t	At	Ft		|E|=|At-Ft|		MAD			0.1				
3	1	130	130.0	=B3	0.00	=ABS(B3-C3)	0.00	=AVERAGE($E$3:E3)						
4	2	150	130.0	=(1-$C$1)*C3+$C$1*B3	20.00	=ABS(B4-C4)	10.00	=AVERAGE($E$3:E4)			39.05			
5	3	80	138.0	=(1-$C$1)*C4+$C$1*B4	58.00	=ABS(B5-C5)	26.00	=AVERAGE($E$3:E5)		0	34.9167		=J1	
6	4	190	114.8	=(1-$C$1)*C5+$C$1*B5	75.20	=ABS(B6-C6)	38.30	=AVERAGE($E$3:E6)		0.1	36.3043		=J5+$J$2	
7	5	140	144.9	=(1-$C$1)*C6+$C$1*B6	4.88	=ABS(B7-C7)	31.62	=AVERAGE($E$3:E7)		0.2	37.2378		=J6+$J$2	
8	6	150	142.9	=(1-$C$1)*C7+$C$1*B7	7.07	=ABS(B8-C8)	27.53	=AVERAGE($E$3:E8)		0.3	37.9191		=J7+$J$2	
9	7	120	145.8	=(1-$C$1)*C8+$C$1*B8	25.76	=ABS(B9-C9)	27.27	=AVERAGE($E$3:E9)		0.4	39.0548		=J8+$J$2	
10	8	140	135.5	=(1-$C$1)*C9+$C$1*B9	4.55	=ABS(B10-C10)	24.43	=AVERAGE($E$3:E10)		0.5	40.181		=J9+$J$2	
11	9	100	137.3	=(1-$C$1)*C10+$C$1*B10	37.27	=ABS(B11-C11)	25.86	=AVERAGE($E$3:E11)		0.6	41.3821		=J10+$J$2	
12	10	60	122.4	=(1-$C$1)*C11+$C$1*B11	62.36	=ABS(B12-C12)	29.51	=AVERAGE($E$3:E12)		0.7	42.7226		=J11+$J$2	
13	11	11	97.4	=(1-$C$1)*C12+$C$1*B12	86.42	=ABS(B13-C13)	34.68	=AVERAGE($E$3:E13)		0.8	44.2571		=J12+$J$2	
14	12	150	62.9	=(1-$C$1)*C13+$C$1*B13	87.15	=ABS(B14-C14)	39.05	=AVERAGE($E$3:E14)		0.9	46.0448		=J13+$J$2	
15	13		97.7	=(1-$C$1)*C14+$C$1*B14						1	48.1667		=J14+$J$2	
16	Forecast for next period:				μ=	97.7	σ=	48.8						
17						=C15		=1.25*G14						



If you do not see the Analysis tool pack or solver in your data tab, you may follow the steps below to access them.